5. CALCULUS

5.1 Defining functions

To enter the function $f(x) = x^2 - 3x + 5$, type

```
> f:= x -> x^2 - 3*x + 5;
```

$$f := x \to x^2 - 3x + 5$$

The arrow symbol is entered by typing the *minus* key, "−" immediately followed by the *greater than* key, ">". We compute $f(2)$.

```
> f(2);
```

$$3$$

Thus, in MAPLE the syntax for creating a function $f(x)$ is `f := x -> expr`, where `expr` is some expression involving x. Functions in more than one variable are defined in the same way.

```
> g := (x,y) -> x*y/(1+x^2+y^2);
```

$$g := (x, y) \to \frac{xy}{1 + x^2 + y^2}$$

We defined the function

$$g(x, y) = \frac{xy}{1 + x^2 + y^2}.$$

Try simplifying $g(\sin t, \cos t)$

```
> g(sin(t),cos(t));
> simplify(%);
```

To convert an expression into a function, we use the `unapply` function.

```
> q := Z^5+3*Z^4-12*Z^3-35*Z^2
  +42*Z+119:
> h := unapply(q,Z);
```

$$h := Z \to Z^5 + 3Z^4 - 12Z^3 - 35Z^2 + 42Z + 119$$

In Sections 3.2 and 3.3 we came across the quintic polynomial q above. Here q is an expression involving Z. To convert q into the function $h(Z)$, we used the command `unapply(q,Z)`. Now we are free to play with the function h.

```
> H := x -> evalf( h(x), 4):
```

$$H := x \to \text{evalf}(\ h(x),\ 4)$$

```
> X := [seq(evalf(-4+i/10,4),i=0..10)];
```

$$X := [-4., -3.900, -3.800, -3.700, -3.600, \\ -3.500, -3.400, -3.300, -3.200, -3.100, -3.]$$

```
> Y := map(H,X);
```

$$Y := [-97., -73.7, -54.5, -39.0, -26.6, -17.1, \\ -10.4, -5.1, -1.4, .6, 2.]$$

The function $H(x)$ computes $h(x)$ to 4 digits. Then we used `seq` and `map` to produce the lists X and Y, which give a table of x and y values for the function $y = h(x)$.

5.2 Composition of functions

In MAPLE, `@` is the function composition operator. If f and g are functions, then the composition of f and g, $f \circ g(x) = f(g(x))$, is given by `(f@g)(x)`.

```
> (sin@cos)(x);
```

$$\sin(\cos(x));$$

```
> f := x -> x^2:
> g := x -> sqrt(1-x):
> (f@g)(x);
```

$$1 - x$$

```
> (g@f)(x);
```

$$\sqrt{1 - x^2}$$

`@@` gives repeated composition, so that `(f@@2)(x)` gives $f(f(x))$ and `(f@@3)(x)` gives $f(f(f(x)))$. For certain functions known to MAPLE, `f@@(-1)(x)` gives the inverse function $f^{-1}(x)$.

5.3 Summation and product

In MAPLE, the syntax for the sum

$$\sum_{i=1}^{n} f(i) = f(1) + f(2) + \cdots + f(n)$$

is `Sum(f(i),i=1..n)` and `sum(f(i),i=1..n)`.

```
> f := 'f':
> Sum(f(i),i=1..n);
```

$$\sum_{i=1}^{n} f(i)$$

```
> Sum(i^2,i=1..10);
```

$$\sum_{i=1}^{10} i^2$$

```
> sum(i^2,i=1..10);
```

$$385$$

Notice that the difference between `sum` and `Sum` is that in `sum`, the sum is evaluated, but that in `Sum`, it is not. It is recommended that you get into the habit of using `Sum` to first check for typos and then use `value` to evaluate the sum. In our previous session we found

$$\sum_{i=1}^{10} i^2 = 1 + 4 + 9 + \cdots + 100 = 385.$$

This time we will use `Sum` and `value`.

```
> Sum(i^2,i=1..10);
```

$$\sum_{i=1}^{10} i^2$$

```
> value(%);
```

$$385$$

```
> sum(i^2,i=1..n);
```

$$1/3\,(n+1)^3 - 1/2\,(n+1)^2 + 1/6\,n + 1/6$$

```
> factor(%);
```

$$1/6\,n\,(n+1)\,(2n+1)$$

Notice that MAPLE knows certain summation formulas such as

$$\sum_{i=1}^{n} i^2 = \frac{1}{6}n(n+1)(2n+1).$$

In MAPLE, the syntax for the product

$$\prod_{i=1}^{n} f(i) = f(1) \cdot f(2) \cdots f(n)$$

is `Product(f(i),i=1..n)`.

```
> f := 'f': q := 'q':
> Product(f(i),i=1..n);
```

$$\prod_{i=1}^{n} f(i)$$

```
> Product(1-q^i,i=1..5);
```

$$\prod_{i=1}^{5} 1 - q^i$$

```
> value(%);
```

$$(1-q)(1-q^2)(1-q^3)(1-q^4)(1-q^5)$$

```
> expand(%);
```

$$-q^{15} + q^{14} + q^{13} - q^{10} - q^9 - q^8 + q^7 + q^6 + q^5 - q^2 - q + 1$$

As with `sum` and `Sum`, for `product`, the product is evaluated, but with `Product`, it is not. Note that we could have evaluated the product $\prod_{i=1}^{5} 1 - q^i$ using `product(1-q^i,i=1..5)`.

A common problem with `sum` and `product` is the following:

```
> i:=2;
```

$$i := 2$$

```
> sum(i^3,i=1..5);
Error, (in sum) summation variable previously assigned,
second argument evaluates to, 2=1 .. 5
```

The problem occurred in `sum` since i had already been assigned the value 2. There are two ways around this problem. One way is to restore the variable status of i by typing `i := 'i'`. The second way is to replace `i` by `'i'` in the sum.

```
> sum('i'^3,'i'=1..5);
```

$$225$$

5.4 Limits

Naturally, there are two forms of the MAPLE limit function: `Limit` and `limit`. These are analogous to `sum` and `Sum`, etc.

The syntax for computing the limit of $f(x)$ as $x \to a$ is `Limit(f(x), x=a); value(%)`. The `Limit` command displays the limit so that it can be checked for typos and then the `value` command computes the limit. To compute the limit

$$\lim_{x\to 2} \frac{x^2-4}{x-2}$$

we type

```
> Limit((x^2-4)/(x-2),x=2); value(%);
```

$$\lim_{x\to 2} \frac{x^2-4}{x-2}$$

$$4$$

Thus, we see that

$$\lim_{x\to 2} \frac{x^2-4}{x-2} = 4,$$

which can be verified easily with paper and pencil. Alternatively, by typing `limit((x^2-4)/(x-2),x=2)`, we could have found the limit in one step.

Left and right limits can also be calculated as well as limits where x approaches infinity. Try

```
> f:=(x^2-4)/(x^2-5*x+6);
> Limit(f,x=3,right); value(%);
> Limit(f,x=infinity); value(%);
```

5.5 Differentiation

MAPLE can easily find the derivatives of functions of one or several variables. The syntax for differentiating $f(x)$ is `diff(f(x),x)`.

```
> f := sqrt(1 - x^2);
```

$$f := \sqrt{1-x^2}$$

```
> diff(f,x);
```

$$-\frac{x}{\sqrt{1-x^2}}$$

```
> g := z -> z^2*exp(z) + sin(log(z)):
> diff(g(x),x);
```

$$2x\, e^x + x^2\, e^x + \frac{\cos(\ln(x))}{x}$$

The second derivative is given by typing `diff(f(x),x,x)`. For the nth derivative, use `diff(f(x),x$n)`. Use MAPLE to show that

$$\begin{aligned}\frac{d^5 \tan x}{dx^5} &= 136 \tan^2 x + 240 \tan^4 x \\ &\quad + 120 \tan^6 x + 16.\end{aligned}$$

In MAPLE, partial derivatives are computed using `diff`.

```
> z := exp(x*y)*(1+sqrt(x^2+3*y^2-x));
```

$$z := e^{xy}\left(1+\sqrt{x^2+3\,y^2-x}\right)$$

```
> diff(z,x);
```

$$ye^{xy}\left(1+\sqrt{x^2+3\,y^2-x}\right) + \frac{e^{xy}\,(2\,x-1)}{2\,\sqrt{x^2+3\,y^2-x}}$$

```
> normal(diff(z,x,y)-diff(z,y,x));
```

$$0$$

The syntax for $\frac{\partial z}{\partial x}$ is `diff(z,x)` and for $\frac{\partial^2 z}{\partial y \partial x}$ is `diff(z,x,y)`. For

$$z = e^{xy}\left(1 + \sqrt{x^2 + 3\,y^2 - x}\right)$$

we found that

$$\frac{\partial z}{\partial x} = y e^{xy}\left(1 + \sqrt{x^2 + 3\,y^2 - x}\right) + \frac{e^{xy}\,(2\,x - 1)}{2\,\sqrt{x^2 + 3\,y^2 - x}},$$

and

$$\frac{\partial^2 z}{\partial y \partial x} = \frac{\partial^2 z}{\partial x \partial x}.$$

MAPLE also has the differential operator D. If f is a differentiable function of one variable, then Df is the derivative f'. We calculate $g'(x)$ for our function g above.

```
> g := z -> z^2*exp(z) + sin(z);
```

$$g := z \to z^2\, e^z + \sin(z)$$

```
> D(g);
```

$$z \to 2z\, e^z + z^2 e^z + \cos(z)$$

5.6 Extrema

MAPLE is able to find the minimum and maximum values of certain functions of one or several variables with zero or more constraints. There are three possible approaches: (1) using the built-in functions `maximize` and `minimize`, (2) using the `extrema` function, and (3) using the *simplex* package (for linear functions). Here we will describe (1) and (2). See `?simplex` for (3).

The functions `maximize` and `minimize` can find the maximum and minimum values of a function of one or several variables. There is also an option for restricting some of the variables to certain intervals. It is advised that this facility be used with care, especially in earlier versions of MAPLE.

We can find the maximum value of the function $f(x)$ using `maximize(f(x))`. The command `maximize(f(x), x=a..b)` gives the maximum of the function, with x restricted to the interval $[a, b]$.

```
> maximize(sin(x));
```

$$1$$

```
> maximize(sin(x)+cos(x));
```

$$\mathit{maximize}(\sin(x) + \cos(x))$$

```
> maximize(x^2-5*x+1,x=0..3);
```

$$1$$

```
> maximize(sin(x),x=0..1);
```

$$\sin(1)$$

```
> maximize(sin(x)+cos(x),x=0..1);
```

$$\sqrt{2}$$

```
> maximize(sin(x)+cos(x),x=0..1/2);
```

$$\sin(1/2) + \cos(1/2)$$

MAPLE was able to find the correct maximum value of $\sin x$, but was unable to compute the maximum for the function $\sin x + \cos x$, although it was able to do so correctly when x was restricted to an interval. For $0 \leq x \leq 3$, the maximum value of $x^2 - 5x + 1$ was found to be 1.

Warning: In MAPLE V Release 5 (and earlier versions), the `maximize` function has a different syntax. In these earlier versions, the correct syntax has the form `maximize(f(x), {x},{x=a..b})`. Bugs in earlier versions have been eliminated in MAPLE 6. For instance, in MAPLE V, the call `maximize(sin(x), {x}, {x=0..1})` will return a value of 1 when the correct value for the maximum of $\sin x$ on the interval $[0, 1]$ is $\sin 1$. In MAPLE 6, the correct value is returned.

To find the minimum value of a function, use the command `minimize` whose syntax is analogous to that of `maximize`. MAPLE can also handle functions of more than one variable.

```
> minimize(x^2+y^2);
```

$$0$$

```
> minimize(x^2+y^2,x);
```

$$y^2$$

We found the minimum value of $x^2 + y^2$ to be 0. The function `minimize(x^2 + y^2,x)` found the minimum value of the function $x^2 + y^2$, considered as a function of x with y fixed.

The second method involves using the function `extrema`, which is able to find the minimum and maximum values of algebraic functions of one or several variables, subject to 0 or more constraints. It returns a set of possible maximum and minimum values, with the option of returning a possible set of points where these values occur. The syntax for the function is `extrema(f,{g1,g2, ... ,gn},{x1,x2, ... ,xm},'s')`. Here, f is the function. The constraints are $g_1 = 0$, $g_2 = 0, \ldots$, $g_n = 0$. The variables are $x_1, x_2, \ldots, x_m$, and s is the unevaluated variable for holding the set of possible points where the extrema occur.

Warning: In MAPLE V Release 5 (and earlier versions), `extrema` is a misc library function, which must be read into our MAPLE session with `readlib(extrema)`.

```
> readlib(extrema):
```

The `readlib` function is obsolete in MAPLE 6 and can be omitted.

```
> f := 2*x^2 + y + y^2;
```

$$f := 2x^2 + y + y^2$$

```
> g := x^2 + y^2 - 1;
```

$$g := x^2 + y^2 - 1$$

```
> extrema(f,{g},{x,y},'s');
```

$$\{0, 9/4\}$$

```
> s;
```

$$\{\{x = 0, y = 1\}, \{x = 0, y = -1\}\},$$
$$\{\{y = 1/2, x = 1/2\mathrm{RootOf}(\_Z^2 - 3)\}\}$$

```
> simplify(subs(s[1],f));
```

$$0$$

```
> simplify(subs(s[2],f));
```

$$2$$

```
> simplify(subs(s[3],f));
```

$$9/4$$

By using the command `extrema(f,{g},{x,y}, 's')`, we found that the extreme values of $f(x,y) = 2x^2 + y + y^2$ (subject to the constraint $x^2 + y^2 = 1$) are 0 and 9/4. The set of possible points where the extrema occured was assigned to the variable s. Using `simplify` and `subs`, we substituted each set of points into f. In this way, we found that the minimum value 0 occurs at $x = 0, y = -1$ and the maximum value 9/4 occurs at $x = \pm\sqrt{3}/2, y = 1/2$.

5.7 Integration

If f is an expression involving x, then the syntax for finding the integral $\int_a^b f(x)\,dx$ is `int(f,x=a..b)`. For the indefinite integral we use `int(f,x)`. There are also the unevaluated forms `Int(f,x=a..b)` and `Int(f,x)`.

```
> Int(x^2/sqrt(1-x^3),x);
```

$$\int \frac{x^2}{\sqrt{1 - x^3}}\,dx$$

```
> value(%);
```

$$-2/3\sqrt{1-x^3}$$

```
> Int(1/x/sqrt(x^2 - 1),x=1..2/sqrt(3));
```

$$\int_1^{2/\sqrt{3}} \frac{1}{x\sqrt{x^2-1}}\,dx$$

```
> value(%);
```

$$\frac{1}{6}\pi$$

MAPLE easily found that

$$\int \frac{x^2}{\sqrt{1-x^3}}\,dx = -\frac{2}{3}\sqrt{1-x^3}$$

and
$$\int_1^{2/\sqrt{3}} \frac{1}{x\sqrt{x^2-1}}\,dx = \frac{\pi}{6}.$$

MAPLE can do improper integrals and multiple integrals in the obvious way. Try finding

$$\int_0^\infty re^{-r^2}\,dr$$

by typing `int(r*exp(-r^2),r=0..infinity)`. Try evaluating the double integral

$$\int\int y\,\sin(2x+3y^2)\,dx\,dy$$

by first integrating with respect to x and then with respect to y.

If MAPLE does not know the value of a definite integral, try `evalf`.

```
> Int(sqrt(1+x^6),x=0..1);
```

$$\int_0^1 \sqrt{1+x^6}\,dx$$

```
> value(%);
```

$$\int_0^1 \sqrt{1+x^6}\,dx$$

```
> evalf(%);
```

$$1.064088379$$

Although MAPLE was unable to evaluate the integral, it was able to find the approximation

$$\int_0^1 \sqrt{1+x^6}\,dx \approx 1.064088379.$$

5.7.1 Techniques of integration

MAPLE knows some standard techniques of integration. These are in the *student* package and are loaded with the command `with(student)`.

5.7.1.1 Substitution

In MAPLE, to do integration by substitution, we use the `changevar` command. The syntax is `changevar(f(u)=h(x),`*integral*`,u)` where *integral* is an integral in the variable x, $f(u) = h(x)$ is the substitution, and u is the new variable in the integral.

```
> with(student):
> G:=Int(x^4/sqrt(1-x^10),x);
```

$$\int \frac{x^4}{\sqrt{1-x^{10}}}\,dx$$

```
> changevar(u=x^5,G,u);
```

$$\int 1/5\,\frac{1}{\sqrt{1-u^2}}\,du$$

```
> G2 := value(%);
```

$$1/5\,\arcsin(u)$$

```
> subs(u=x^5,G2);
```

$$1/5\,\arcsin(x^5)$$

```
> diff(%,x);
```

$$\frac{x^4}{\sqrt{1-x^{10}}}$$

Using `changevar` with the substitution $u = x^5$, we found

$$\begin{aligned}\int \frac{x^4}{\sqrt{1-x^{10}}}\,dx &= \frac{1}{5}\int \frac{1}{\sqrt{1-u^2}}\,du\\ &= \sin^{-1} u\\ &= \sin^{-1}(x^5)\end{aligned}$$

```
> G:=Int((3*x^2+1)/sqrt((1-x-x^3)*(1+x+x^3)),x);
```

$$G := \int \frac{3\,x^2+1}{\sqrt{(1-x-x^3)\,(1+x+x^3)}}\,dx$$

```
> value(G);
```

$$\int \frac{3\,x^2+1}{\sqrt{(1-x-x^3)\,(1+x+x^3)}}\,dx$$

Although MAPLE was unable to evaluate the integral above, you should be able to help it along by using `changevar` and the substitution $u = x + x^3$.

```
> radsimp(changevar(u=x+x^3,G,u));
```

5.7.1.2 Integration by parts

To do integration by parts, we use the command `intparts`. The syntax is `intparts(`*integral*`, u)` where u is as usual in the formula

$$\int u\,dv = uv - \int v\,du.$$

```
> with(student):
> Int(x*cos(3*x),x);
```

$$\int x\,\cos 3x\,dx$$

```
> intparts(%,x);
```

$$1/3\,x\,\sin(3x) - \int 1/3\,\sin(3x)\,dx$$

```
> value(%);
```

$$1/3\,x\,\sin(3x) + 1/9\,\cos(3x)$$

Thus MAPLE has helped us by providing the details of the evaluation of the integral by parts:

$$\begin{aligned}\int x\,\cos 3x\,dx &= 1/3\,x\,\sin 3x - \int 1/3\,\sin 3x\,dx\\ &= 1/3\,x\,\sin 3x + 1/9\,\cos 3x.\end{aligned}$$

5.7.1.3 Partial fractions

The command for finding the partial fraction decomposition of a rational function *ratfunc* (in the variable x) is `convert(`*ratfunc*`,parfrac,x)`. As an example, we use MAPLE to find the integral

$$\int \frac{4x^4 + 9x^3 + 12x^2 + 9x + 4}{(x+1)(x^2+x+1)^2}\,dx.$$

```
> rat := (4*x^4+9*x^3+12*x^2+9*x+4)
    /(x + 1)/(x^2 + x + 1)^2:
> convert(rat,parfrac,x);
```

$$\frac{2}{x+1} + \frac{1+2x}{x^2+x+1} + \frac{1}{(x^2+x+1)^2}$$

```
> int(%,x);
```

$$2\ln(x+1)+\ln(x^2+x+1)+\frac{1}{3}\,\frac{2x+1}{x^2+x+1}$$
$$+\frac{4}{9}\sqrt{3}\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right)$$

5.8 Taylor and series expansions

The command to find the first n terms of the Taylor series expansion for $f(x)$ about the point $x=c$ is `taylor(f(x),x=c,n)`. We compute the first five terms of the Taylor series expansion of $y=(1-x)^{-1/2}$ about $x=0$.

```
> y := 1/sqrt(1-x);
```

$$y:=\frac{1}{\sqrt{1-x}}$$

```
> taylor(y,x=0,5);
```

$$1+\frac{1}{2}x+\frac{3}{8}x^2+\frac{5}{16}x^3+\frac{35}{128}x^4+O\left(x^5\right)$$

To find a specific coefficient in a Taylor series expansion, use `coeff`.

```
> J := product(1-x^'i','i'=1..50):
> taylor(J^3,x=0,20);
```

$$1-3x+5x^3-7x^6+9x^{11}-11x^{15}+O(x^{20})$$

```
> coeff(%,x,15);
```

$$-11$$

To convert a *series* into a polynomial, try `convert(`*series*`, polynom)`. Also, see `?series`.

5.9 The *student* package

The *student* package contains many functions to help the calculus student solve problems step-by-step. In Section 5.7.1 we used the *student* package functions `changevar`, `intparts` to do some integration problems. The package includes the following functions:

D	Diff	Doubleint	Int	Limit
Lineint	Point	Product	Sum	Tripleint
changevar	combine	completesquare	distance	equate
extrema	integrand	intercept	intparts	isolate
leftbox	leftsum	makeproc	maximize	middlebox
middlesum	midpoint	minimize	powsubs	rightbox
rightsum	showtangent	simpson	slope	summand
trapezoid				

We give a brief description of the main functions.

Doubleint

Calculates double integrals. `Doubleint(f,x,y)` is equivalent to `int(int(f,x),y)` and `Doubleint(f,x=a..b,y=c..d)` is equivalent to `int(int(f,x=a..b),y=c..d)`. Also see Section 10.6.1.

Lineint

Calculates line integrals. Suppose a curve $\mathcal{C}$ is parameterized by $x = x(t)$, $y = y(t)$ $(a \le t \le b)$, and $f(x,y)$ is a function defined on $\mathcal{C}$. Let $\vec{r}(t) = x(t)\vec{i} + y(t)\vec{j}$. The line integral

$$\int_{\mathcal{C}} f(x,y)\,ds = \int_a^b f(x(t),y(t))\,||\vec{r}\,'(t)||\,dt$$

is given in MAPLE by `Lineint(f(x,y),x,y,t=a..b)`. Also see Section 10.8.

Tripleint

Calculates triple integrals and is analogous to `Doubleint`. Also see Section 10.6.2.

completesquare

`completesquare` is used to complete the square.

```
> with(student):
> p := x^2 + 6*x + 13;
```

$$x^2 + 6\,x + 13$$

```
> completesquare(p);
```

$$(x+3)^2 + 4$$

```
> q := x^2 + 10*x + 2*y^2 + 12*y + 12;
```

$$x^2 + 10\,x + 2\,y^2 + 12\,y + 12$$

```
> completesquare(q);
Error, (in completesquare) unable to choose indeterminate
> completesquare(q,x);
```

$$(x+5)^2 - 13 + 2\,y^2 + 12\,y$$

```
> completesquare(%,y);
```

$$2\,(y+3)^2 - 31 + (x+5)^2$$

We found that

$$x^2 + 6\,x + 13 = (x+3)^2 + 4,$$
$$x^2 + 10\,x + 2\,y^2 + 12\,y + 12 = 2\,(y+3)^2 - 31 + (x+5)^2\,.$$

distance

Finds the distance between two points in one, two, or three dimensions.

```
> with(student):
> distance(-3,5);
```

$$8$$

```
> distance([1,2],[-3,4]);
```

$$2\sqrt{5}$$

We see that the distance between the two real numbers -3 and 5 is $|-3-5| = 8$ and that the distance between the points $(1, 2)$ and $(-3, 4)$ is $2\sqrt{5}$.

equate

Generates a set of equations.

```
> with(student):
> equate(x,y);
```

$$\{x = y\}$$

```
> equate([x+y,x-y],[3,-1]);
```

$$\{x - y = -1, x + y = 3\}$$

```
> solve(%);
```

$$\{y = 2, x = 1\}$$

integrand

Extracts the integrand from an inert MAPLE integral.

```
> with(student):
> F := Int(sin(x^3*y),x=0..2*Pi);
```

$$F := \int_0^{2\pi} \sin(x^3 y) dx$$

```
> integrand(F);
```

$$\sin(x^3 y)$$

intercept

Computes the x-intercept as well as the intersection point of two curves.

```
> with(student):
> intercept(y=5*x-3);
```

$$\{y = -3, x = 0\}$$

```
> intercept(y=x^2+3*x-20,y=2*x^2+x-23);
```

$$\{x = -1, y = -22\}, \{x = 3, y = -2\}$$

We see that the x-intercept of the line $y = 5x - 3$ is the point $(0, -3)$ and that the curves $y = x^2 + 3x - 20$, $y = 2x^2 + x - 23$ have two intersection points $(-1, -22)$ and $(3, -2)$.

leftbox

Gives a graphical representation of a certain Riemann sum. The command `leftbox(f(x), x=a..b, n)` graphs $f(x)$ on the interval $[a, b]$ as well as n rectangles whose area approximates the definite integral. The left corner of each rectangle is a point on the graph of $y = f(x)$. We use `leftbox` to give a graphical approximation for the integral $\int_0^\pi \sin x^2\, dx$:

```
> with(student):
> leftbox(sin(x^2), x=0..Pi, 6, shading=green);
```



Figure 5.1 Rectangles representing a Riemann sum.

MAPLE's plotting functions are treated in detail in the next chapter. Related functions are `rightbox` and `middlebox`.

leftsum

`leftsum` is the Riemann sum that corresponds to `leftbox`. We compute the Riemann sum, which corresponds to the areas of the rectangles in our previous example.

```
> with(student):
> leftsum(sin(x^2), x=0..Pi, 6);
```

$$1/6\,\pi \sum_{i=0}^{5} \sin(1/36\, i^2 \pi^2)$$

```
> value(%);
```

$$1/6\,\pi \left(\sin(1/36\,\pi^2) + \sin(\frac{1}{9}\,\pi^2) + \sin(\frac{1}{4}\,\pi^2) + \sin(\frac{4}{9}\,\pi^2) + \sin(\frac{25}{36}\,\pi^2)\right)$$

```
> evalf(%);
```

$$0.7212750238$$

```
> evalf(int(sin(x^2),x=0..Pi));
```

$$0.7726517130$$

The required Riemann sum is

$$\frac{\pi}{6}\sum_{i=0}^{5}\sin(i^2\pi^2/36) \approx .7212750238,$$

which is an approximation of the integral

$$\int_0^\pi \sin x^2\,dx = 0.772651713\cdots.$$

Related functions are `rightsum` and `middlesum`.

<u>`makeproc`</u>

The `makeproc` is used for defining functions and takes three forms. If *expr* is an expression in the variable x, then `makeproc` converts the expression into a function of x.

```
> with(student):
> y := x^2 + x - 3;
```

$$y := x^2 + x - 3$$

```
> f := makeproc(y,x);
```

$$f := x \mapsto x^2 + x - 3$$

```
> f(x);
```

$$x^2 + x - 3$$

We converted the expression $x^2 + x - 3$ into a function of x. Also see `?unapply`. To find the linear function whose graph passes through the two points (a, b), (c, d), use the command `makeproc([a,b],[c,d])`.

```
> with(student):
> f := makeproc([-1,1],[3,7]);
```

$$f := x \mapsto \frac{3}{2}x + \frac{5}{2}$$

```
> f(-1), f(3);
```

$$1, 7$$

We see that $y = \frac{3}{2}x + \frac{5}{2}$ is the line that passes through the two points $(-1, 1)$ and $(3, 7)$. To find the linear function whose graph passes through (a, b) and has slope m, use the command `makeproc([a,b],'slope'=m)`.

```
> with(student):
> f := makeproc([2,5],'slope'=3);
```

$$f := x \mapsto 3x - 1$$

```
> f(2);
```

$$5$$

```
> diff(f(x),x);
```

$$3$$

We see that $y = 3x - 1$ is the line with slope 3 that passes through the point $(2, 5)$.

midpoint

To find the midpoint of the line segment joining the two points (a, b), (c, d), use the command `midpoint([a,b],[c,d])`.

```
> with(student):
> midpoint([2,3],[5,7]);
```

$$[\frac{7}{2}, 5]$$

We see that the midpoint of the segment joining the points $(2, 3)$, $(5, 7)$ is the point $(7/2, 5)$.

powsubs

The `powsubs` function behaves like the `subs` function. See `?powsubs` and `?subs` for more information.

showtangent($f(x)$, x=a)

Produces a graph of the function $y = f(x)$ near $x = a$ together with the tangent that passes through the point $(a, f(a))$. We graph the tangent to the curve $y = \sin x$ at $x = 2\pi/5$ together with the curve.

```
> with(student):
> showtangent(sin(x),x=Pi/4);
```



Figure 5.2 The function $y = \sin x$ and the tangent at $x = 2\pi/5$.

simpson

Computes an approximation to a definite integral using Simpson's rule. The call `simpson(`$f(x)$`,`x`,`n`)` finds an approximation to the definite integral $\int_a^b f(x)\,dx$ using n subdivisions. We use Simpson's rule with $n = 12$ to find an approximation to $\int_0^1 \frac{1}{\sqrt{1+x^4}}\,dx$:

```
> with(student):
> simpson(1/sqrt(1+x^4),x=0..1,12):
> value(%):
> app := evalf(%);
```

$$app := 0.9270384891$$

```
> xval := evalf(int(1/sqrt(1+x^4),x=0..1));
```

$$xval := 0.9270373385$$

```
> abs(app-xval);
```

$$.11506\,10^{-5}$$

We found that

$$\int_0^1 \frac{1}{\sqrt{1+x^4}}\,dx \approx 0.9270384891,$$

and the error $< 10^{-5}$.

slope

Gives the slope of a line.

```
> with(student):
> slope(y=2*x-5);
```

```
> slope(2*y+12=3*x);
Error, (in slope) use slope(y=f(x)), or slope(f(x,y)=g(x,y),y(x))
> slope(2*y+12=3*x,y(x));
```

$$3/2$$

```
> slope([12,5],[3,7]);
```

$$-\frac{2}{9}$$

We found that the slope of the line $y = 2x - 5$ is 2. To find the slope of the line $2y + 12 = 3x$, we need to tell MAPLE that y is the dependent variable. Using the call `slope(2*y+12=3*x,y(x))`, we found the slope to be $3/2$. The call `slope([12,5],[3,7])` gives the slope of the line segment joining the points $(12, 5)$ and $(3, 7)$.

summand

Gives the summand in a sum.

```
> with(student):
> z3 := Sum(1/n^3,n=1..infinity);
```

$$z3 := \sum_{n=1}^{\infty} \frac{1}{n^3}$$

```
> summand(z3);
```

$$\frac{1}{n^3}$$

trapezoid

Uses the trapezoidal rule to compute an approximation to a definite integral. The call `trapezoid(`$f(x)$`,`x`,`n`)` finds an approximation to the definite integral $\int_a^b f(x)\,dx$ using n subdivisions. We use the trapezoidal rule with $n = 12$ to find an approximation to $\int_0^1 \frac{1}{\sqrt{1+x^4}}\,dx$:

```
> with(student):
> trapezoid(1/sqrt(1+x^4),x=0..1,12):
> value(%):
> app := evalf(%);
```

$$0.9266278484$$

```
> xval := evalf(int(1/sqrt(1+x^4),x=0..1));
```

$$0.9270373385$$

```
> abs(app-xval);
```

$$0.0004094901$$

This time we found that

$$\int_0^1 \frac{1}{\sqrt{1+x^4}}\,dx \approx 0.9266278484,$$

and the error $< 10^{-3}$. The approximation found earlier using Simpson's rule was better.